
ADTs, Functions, Induction

Andrei Arusoaie

Foundations of Programming Languages

Contents

1 ADTs, Functions, Induction 2
1.1 Algebraic data types and functions . . . . . . . . . . . . . . . . . . . . . . . . . . 2

1.1.1 Enumerated types . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2
1.1.2 Inductive definitions . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3
1.1.3 Recursive functions . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4
1.1.4 Bool and Nat in Lean 4 . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5

1.2 Proofs . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5
1.2.1 Simplification . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5
1.2.2 Rewriting . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5
1.2.3 Case analysis . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6

2 Induction 6

1



1 ADTs, Functions, Induction

In these notes we use a proof assistant, Lean 4, as a framework for defining programming-
language semantics. Within this framework we can execute programs written in the languages
we define, and we can prove properties about those languages and programs. Lean 4 supports
a functional programming style for defining algebraic data types and higher-order functions,
together with pattern matching and polymorphism. Many of these concepts are presented from
different perspectives in several seminal works (e.g., [3, 2, 4, 1]).

1.1 Algebraic data types and functions

Algebraic data types are data types formed using other types. The term is typically associated
with functional programming and type theory.

1.1.1 Enumerated types

The simplest algebraic data type is defined by enumerating its values. Here is a simple enumer-
ation, a type called Season:

inductive Season where

| Winter

| Spring

| Summer

| Fall

The keyword inductive introduces the new type Season. Its values — Winter, Spring, Summer,
Fall — are also called constructors. By default a constructor is referred to through its type,
e.g. Season.Winter, since another type could also have a constructor named Winter. To avoid
this qualification, we open the namespace of Season and then refer to its constructors directly:

open Season

#check Winter

The open command brings all constructors of Season into the current namespace, so Winter is
understood as Season.Winter without the prefix.

Exercise 1.1. Check the type of the other values of Season.

We can write functions over the values of Season, e.g., a function returning the next season:

def next_season (s : Season) : Season :=

match s with

| Season.Winter => Season.Spring

| Season.Spring => Season.Summer

| Season.Summer => Season.Fall

| Season.Fall => Season.Winter

The def keyword introduces a new function, next season, taking an argument s of type Season
and returning a value of type Season:

#check next_season

-- next_season : Season -> Season

Here the types are declared explicitly; Lean 4 can often infer them automatically (type in-
ference), but explicit types improve readability. The body uses pattern matching : the match

construct inspects the constructor of s and returns the value from the matching branch. Testing
the function:

#eval next_season Season.Winter

-- Season.Spring

Exercise 1.2. Evaluate next season on the other possible inputs.

1.1.2 Inductive definitions

Inductive definitions [2] are a central mathematical instrument in the theoretical study of pro-
gramming languages. An inductive definition is a set of inference rules of the shape

I1 . . . In
C

name,

where I1, . . . , In are premises, C is the conclusion, and name is the rule name (optionally with
a side condition). The rule states that the premises are sufficient for the conclusion: to show
C holds, it suffices to show I1, . . . , In hold. If n = 0 (no premises), the rule is called an axiom,
and, instead we use a central dot “·” to denote the lack of premises.

The following two rules form an inductive definition of the natural numbers:

·
O ∈ N

zero
n ∈ N
S n ∈ N

succ

This specifies that N contains O (via the axiom zero) and S n whenever n ∈ N (via succ). This
is the Peano definition of the naturals. In Lean 4:

inductive nat where

| O : nat

| S : nat -> nat

nat is built from two constructors: O is a constant value, exactly as Winter was a constant
value of Season; S is more interesting, since it can construct infinitely many values of type nat:
since O has type nat, so does S O, and so does S (S O), and so on — we have just written down
a representation of the natural numbers. The symbols O and S are just names with no built-in
meaning; the usual reading is that O is 0, S O is 1, S (S O) is 2, and so on.

We can inspect the definition and build values. After opening the nat namespace, its
constructors are available directly:

#print nat

open nat

#check S (S O)

#check O.S.S

The last line uses dot notation: O.S stands for nat.S O (that is, S O), and O.S.S for nat.S

(nat.S O) — so O.S.S and S (S O) denote the very same value. How does Lean 4 know that
S (S O) has type nat? It computes a derivation of this fact, using zero and succ:

·
O ∈ N

zero

S O ∈ N
succ

S (S O) ∈ N
succ

Derivations are finite compositions of inference rules; reading the derivation above bottom-to-
top, the rules composed are succ, succ, and zero. Derivations must end in axioms — otherwise
they would not be finite, and the term “derivation” would not even apply. Notice that the rule

n ∈ N
S n ∈ N

succ

is used twice, for two different values of n (n = O and n = S O): this is why we call succ a rule
scheme — it can be instantiated for infinitely many values of the metavariable n.

Exercise 1.3. Consider the following inductive definition of a list of naturals L:

·
Nil ∈ L

nil
n ∈ N l ∈ L

(Cons n l) ∈ L
cons

Write the corresponding Lean 4 code for this definition.

Exercise 1.4. Recall the definition of L above. A possible derivation for (Cons O nil) ∈ L is:

·
O ∈ N

zero
·

Nil ∈ L
nil

(Cons O nil) ∈ L
cons

Write, on paper, a derivation for (Cons (S O) (Cons O nil)) ∈ L.

Exercise 1.5. Provide an inductive definition for binary trees of natural numbers. Give a Lean
4 implementation, and write several non-trivial derivation examples.

1.1.3 Recursive functions

Functions that operate on inductively defined types are typically recursive. We recall the
inductive definition of N alongside the recursive definition of addition (+) on naturals:

·
O ∈ N

zero
n ∈ N
S n ∈ N

succ

O + n = n S m+ n = S (m+ n)

There is a direct correspondence between each inference rule and each case of the recursive
definition of +. Addition has two arguments and is defined recursively on the first : the base
case handles O (constructed via zero), and the recursive case handles S m (constructed via
succ), expressing S m+ n in terms of the smaller computation m+ n.

This idea generalizes. For instance, the length of a list (as defined in the exercise above)
can be defined recursively as:

length(Nil) = O length(Cons n l) = (S O) + length(l)

Exercise 1.6. Write a recursive function returning the last element of a list, returning O if the
last element does not exist.

Exercise 1.7. Write a recursive function computing the number of nodes of a binary tree, using
your inductive definition from the (earlier) Exercise 1.5.

The definition of addition above is encoded in Lean 4 using def, which supports recursive
function definitions:

def plus (m : nat) (n : nat) : nat :=

match m with

| O => n

| S m' => S (plus m' n)

Addition recurses on its first argument m: the base case O returns n, and the step case S m’

returns S (plus m’ n). Testing it:

#eval plus (S O) (S (S O))

Reading O as 0, S O as 1, S (S O) as 2, this evaluates to S (S (S O)), i.e. 1 + 2 = 3, as
expected.

Exercise 1.8. Implement the two recursive functions from the Exercises 1.6 and 1.7 above in
Lean 4, and test them on several non-trivial inputs.

1.1.4 Bool and Nat in Lean 4

The most common types are already built into Lean 4. For example, Bool is defined as:

#print Bool

The #print command shows this definition: two constructors, true and false, corresponding
to the usual truth values.

Exercise 1.9. Use #print to show the definition of Nat in Lean 4, and consult the documen-
tation.

Exercise 1.10. Write a recursive function deciding whether a natural number is even.

Exercise 1.11. Write a recursive function computing the factorial of a natural number.

Exercise 1.12. Write a function returning true iff n1 < n2 for two input naturals n1, n2, and
false when n1 ≥ n2.

1.2 Proofs

In Lean 4 we can state and prove properties of our functions and relations, using a tactic
mechanism: tactics generate a proof object that is automatically checked. This checking process
is called certification. Certification greatly reduces the chance of incomplete or incorrect proofs
— a real risk with paper proofs, where mistakes or overlooked corner cases are common. For
readers familiar with lambda calculus and type systems: Lean 4 tactics generate a term, and
Lean checks that this term has the expected type; if so, the corresponding proposition holds,
via the Curry–Howard correspondence.

1.2.1 Simplification

A first, simple theorem about natural numbers:

-- proof by simplification

theorem zero_plus_m_is_m :

forall m,

plus O m = m :=

by

intro m

simp [plus]

After intro m, the goal to prove is plus O m = m: adding O to any m returns m. The simp

tactic simplifies the goal using the definition of plus, reducing it to m = m, true by reflexivity.

Exercise 1.13. Prove that plus O O = O.

1.2.2 Rewriting

The next theorem states that equal naturals can be safely substituted in an addition:

-- proof by rewriting

theorem plus_zero_eq :

forall m n,

m = n ->

plus O m = plus O n :=

by

intro m n H

rw [H]

The rw (rewrite) tactic replaces m with n in the goal, using the hypothesis H.

Exercise 1.14. Prove that for all naturals m,n, plus m n = plus n n when m = n.

Exercise 1.15. Prove that for all naturals m,n, plus m n = plus m m when m = n.

1.2.3 Case analysis

Simplification and rewriting are not always enough. Case analysis inspects the possible values
of a quantity and shows the target property holds in each case. For instance: adding 1 (that is,
S O) to any natural k can never equal 0. How do we prove this for a nat? By case analysis on
k:

• k = O: then k + S O = O + S O = S O, different from O because they are built from
different constructors.

• k = S k′, for some nat k′: then k + S O = (S k′) + S O = S (k′ + S O), again different
from O for the same reason.

Case analysis proceeds by examining the two constructors, O and S; constructors build distinct
values, so S k′ and O are different by construction, for any k′. In Lean 4:

-- proof by case analysis

theorem plus_with_SO_is_not_O :

forall k,

plus k (S O) ̸= O :=

by

intro k

cases k with

| O => simp [plus]

| S k' => simp [plus]

Exercise 1.16. Fill in the ... below:

theorem plus_with_SO_is_not_O' :

forall k,

plus k (S O) ̸= O :=

by

intro k

cases k with

| O => -- ...

| S k' => -- ...

2 Induction

Proofs by induction are ubiquitous in mathematics, and Lean 4 is no exception. To see why
induction matters, consider proving ∀n. n+ 0 = n using the definitions above:

·
O ∈ N

zero
n ∈ N
S n ∈ N

succ O + n = n (S m) + n = S (m+ n)

This looks obvious, yet it is not trivial to prove: none of the addition rules apply directly to the
left-hand side n+0, since addition is defined by recursion on its first argument, not the second.
Even attempting case analysis on n runs into trouble:

• if n = O: then n + 0 = n becomes O + O = O, which reduces to O = O — true by
reflexivity.

• if n = S n′: then n+0 = n becomes (S n′)+O = (S n′); using the recursive equation, this
becomes S (n′ +O) = S n′. But we have no information yet establishing n′ +O = n′!

This missing fact — n′+O = n′ — is exactly what an inductive hypothesis would give us. Case
analysis alone is not sufficient; we need induction.

An inductive definition is a collection of rules describing how values are constructed. To
prove a property P (n) for n ∈ N, it suffices to show P holds for every way of constructing n:
show P (O) holds, and show P (S n) holds whenever P (n) holds:

·
O ∈ N

zero
n ∈ N
S n ∈ N

succ

·
P (O)

base
P (n)

P (S n)
ind

This is exactly ordinary mathematical induction on the naturals: P (O) is the base case, and
P (n) → P (S n) is the inductive step, in which P (S n) is proved using the inductive hypothesis
P (n). This is the induction principle for naturals:

∀P. P (O) ∧ (∀n. P (n) → P (S n)) → ∀n. P (n)

Returning to ∀n. n+ 0 = n, proved by induction on n:

• base case: P (O): O +O = O, by the first equation of + — true by reflexivity.

• inductive step: assume P (n): n + O = n; prove P (S n): (S n) + O = (S n). Using
the second equation of +, the left side becomes S (n + O); rewriting with the inductive
hypothesis n+O = n gives S n — true by reflexivity.

Induction in Lean 4. Lean 4 automatically generates the induction principle for any induc-
tive definition. For nat, this principle is nat.rec; we can inspect it together with a few basic
checks:

-- induction

#print nat

#check O

#check S O

#check @nat.rec

Lean 4 reports the type of @nat.rec, which is precisely the induction principle for nat:

-- nat.rec : {motive : nat -> Sort u_1} ->

-- motive nat.O ->

-- ((n : nat) -> motive n -> motive (nat.S n)) ->

-- (t : nat) -> motive t

The reserved keyword for induction proofs in Lean 4 is induction; every proof by induction [3]
on nat uses nat.rec under the hood. We can now prove ∀n. n+ 0 = n in Lean 4:

theorem plus_n_O_is_n:

forall n,

plus n O = n :=

by

intro n

induction n with

| O => simp [plus]

| S n' IH => simp [plus]

rw [IH]

In the base case, simp [plus] reduces plus O O = O to a reflexivity. In the step case, simp
[plus] turns the goal into S (plus n’ O) = S n’, and rw [IH] rewrites with the inductive
hypothesis IH : plus n’ O = n’, closing the goal.

Addition on nat is also associative and commutative; both are proved by induction on the
first argument. Associativity is a direct induction:

theorem plus_is_assoc:

forall a b c,

plus (plus a b) c = plus a (plus b c) :=

by intro a

induction a with

| O => simp [plus]

| S a' IHa => intro b c

simp [plus]

rw [IHa]

Commutativity is more delicate: the base case needs plus n O is n as a lemma, and the step
case performs a nested induction on the second argument:

theorem plus_is_comm:

forall m n,

plus m n = plus n m :=

by

intro m

induction m with

| O => intro n

simp [plus]

rw [plus_n_O_is_n]

| S m' IH => intro n

induction n with

| O => simp [plus]

rw [plus_n_O_is_n]

| S n' IH' => simp [plus]

rw [<- IH']

simp [plus]

rw [IH]

simp [plus]

rw [IH]

Exercise 2.1. Recall the list definition from Exercise 1.3. Write a function appending two
lists. State and prove in Lean 4 that appending the empty list to a list l returns l, and that list
append is associative.

Disclaimer

The material in these notes is not claimed to be original; it is heavily based on the four references
below.

